

Power- und Fallzahlanalyse mit SAS 9

Ralf Minkenberg
Repges & Partner GmbH
Neuenhofstr. 114
52078 Aachen
rm@repges-partner.de

Zusammenfassung

Mit der SAS-Version 9 sind über die Prozeduren POWER und GLMPOWER die Möglichkeiten der Power- und Fallzahlanalyse in SAS erheblich erweitert und vereinfacht worden. Neben den Prozeduren sind auch Berechnungen über eine interaktive Oberfläche, der „SAS/STAT Power and Sample Size“-Oberfläche (PSS), auf sehr einfache Weise möglich. Es werden einige typische Situationen und deren Lösung beschrieben und darüber hinaus auch die Möglichkeiten graphischer Ausgaben vorgestellt. Sowohl die Installation und Anwendung von PSS als auch die Syntax der zugrundeliegenden Prozeduren werden erklärt.

Schlüsselworte: Fallzahlplanung, Poweranalyse, PSS, POWER, GLMPOWER

1 Die „SAS/STAT Power and Sample Size“-Oberfläche (PSS)

Auf den für Version 9 vorliegenden Installations-CDs findet sich u.a. die getrennt von SAS zu installierende Anwendung „SAS/STAT Power and Sample Size“ (PSS). Diese ist auf der SAS Mid-Tier Components CD zu finden. Da PSS eine JavaScript-Anwendung ist, muss vor der Verwendung auch ein Webserver vorhanden sein. Dazu kann ein vorhandener Webserver verwendet werden oder auch lokal ein solcher installiert werden. Generell ist es möglich, dass die PSS-Anwendung, der Webserver und das SAS-Programm auf unterschiedlichen Rechnern installiert sind und laufen. Im folgenden soll jedoch nur die Installation aller drei benötigten Anwendungen auf einen lokalen Rechner näher beschrieben werden. PSS selber kann prinzipiell aus jedem Webbrowser aufgerufen werden, wobei Javascript aktiviert sein muss.

1.1 Die Installation von PSS

Mit Hilfe der SAS Mid-Tier Components CD kann die PSS-Anwendung problemlos installiert werden. Nach Einlegen der CD, die unter den Installations-CDs in der Kit Section 803 (Mid-Tier Media) zu finden ist, startet nach Auswahl einer Sprache der sog. SAS Software Navigator automatisch (wenn nicht, muss entsprechend setup.exe von der CD gestartet werden) und es erscheint Abbildung 1.




Abbildung 1: Der SAS Software Navigator

Nach Auswahl des Menüpunktes „SAS/STAT Power and Sample Size“ werden die Voraussetzungen zur Installation beschrieben. Außerdem wird hier ein Link zur Konfiguration und Bedienung des Tomcat Webservers zur Verfügung gestellt. Hier finden sich wichtige Informationen zur Bedienung der PSS-Anwendung und bei auftretenden Problemen. Nach Auswahl des betriebssystemabhängigen Links zur Installation von PSS wird zunächst nachgefragt, wo benutzerabhängige Dateien gespeichert werden sollen und ob PSS lokal oder remote verwendet werden soll. Danach kann der

Tomcat Webserver installiert werden und schließlich die PSS-Anwendung selbst. Nach der erfolgreichen Installation kann PSS gestartet werden, indem nach dem Start des Webservers (standardmäßig zu finden unter Windows im Menüpunkt „Programme/SAS/Power and Sample Size/Tomcat 4.1/Start Tomcat“) in einem Internetbrowser folgende Adresse eingegeben wird: `http://localhost:8080/pss/Power`
Es erscheint der in Abbildung 2 dargestellte Analyse-Bildschirm von PSS, von dem aus alle verfügbaren Fallzahl- und Powerberechnungen gestartet werden können.



Abbildung 2: Der PSS-Analysebildschirm

Die Bedienung von PSS ist übersichtlich und größtenteils selbsterklärend. Die den einzelnen angezeigten Punkten hinterlegten SAS-Prozeduren werden nach Eingabe aller benötigten Parameter in dem (bei lokaler Installation) auf dem Rechner installierten SAS 9 im Hintergrund ausgeführt und die Ergebnisse angezeigt. Leider funktioniert PSS bei installiertem SAS mit deutscher Sprachunterstützung nicht korrekt, es werden dann keine oder offensichtlich fehlerhafte Ergebnisse erzeugt. Es sollte also als Standardsprachunterstützung auf dem Rechner nicht deutsch installiert sein,

denn dann ist eine Benutzung von PSS (zur Zeit) nicht möglich. Bei anderen Sprachen sind entsprechende Probleme bisher nicht bekannt.

1.2 Beispiele zur Benutzung von PSS

Im folgenden soll anhand einiger Beispiele gezeigt werden, wie mit PSS gearbeitet werden kann. Generell stehen folgende Menüpunkte zur Auswahl:

- Means and Location
- Proportions
- Equivalence Testing
- Correlation and Regression
- Analysis of Variance
- Linear Models
- Survival Analysis

Beim Punkte „Means and Location“ können beispielsweise folgende Unterpunkte für Fallzahl- und Poweranalysen Mittelwerte betreffend angewählt werden:

- One-sample t-test
- Paired t-test
- Two-sample t-test
- Confidence interval for one-sample mean
- Confidence interval for paired means
- Confidence interval for two-sample means

1.2.1 Zweistichproben t-Test

Um die Power bei verschiedenen Fallzahlen bei einem unverbundenen t-Test für zwei Stichproben zu berechnen, wählt man den Punkt „Two sample t-test“ unter „Means and Location“ aus.

Wie in Abbildung 3 zu erkennen ist, sind die Punkte „Power“ (alternative Möglichkeiten sind: „Total sample size“, „Sample size per group“), „Two-tailed test“, „Normal“ und Alpha = 0.05 voreingestellt. Weiterhin können nun Mittelwerte für die beiden Gruppen, die gemeinsame Standardabweichung und die Fallzahl je Gruppe eingegeben werden. Hier können jeweils auch mehrere Werte angegeben werden – es werden immer alle möglichen Kombinationen zwischen allen eingegebenen Werten für die Berechnung berücksichtigt. Unter den Auswahlpunkten „Select Alternate Form“ können jeweils auch andere Möglichkeiten der Werteeingabe verwendet werden, bei „Means“ beispielsweise neben „Group1, Group 2“ auch noch „Differences between groups“, „Group 1, Group 2, Null difference“ oder „Differences between

groups, Null difference“. Am Ende der Eingabeseite können noch die optionalen Möglichkeiten der Ergebnisausgabe konfiguriert werden.

Abbildung 3: Eingabemaske für den Zweistichproben t-Test

Durch Drücken des Buttons „Calculate“ erscheint die Meldung „This page has been submitted“ und die Power für die eingegebenen Wertekombinationen wird berechnet. Als beispielhafte Ausgabe für die vorgegebenen Werte $\alpha = 0.05$, Mittelwerte = 10 (Gruppe 1) bzw. 20 (Gruppe 2), Standardabweichung = 10, 15, 20 und Fallzahl je Gruppe = 20, 30 ergibt sich die Ausgabe von Abbildung 4 und die Graphik von Abbildung 5.



Abbildung 4: Powerberechnung für t-Test




Abbildung 5: Graphik: Power für t-Test

Zu allen berechneten Ergebnissen kann dieses jeweils in Textform („Narratives“) ausgegeben werden. Die entsprechende Auswahl kann nach Abbildung 6 vorgenommen werden und mit dem Button „Create Narratives“ gestartet werden. Ein entsprechender Text lautet beispielsweise:

„For a two-sample pooled t-test of a normal mean difference with a two-sided significance level of 0.05, assuming a common standard deviation of 10, a sample size of 20 per group has a power of 0.869 to detect a difference between the means 10 and 20.“

Obs	Select	Index	Sides	Alpha	Mean1	Mean2	Std Dev	N Per Group	Null Diff	Power	Error	Information
1	<input checked="" type="checkbox"/>	1	2	0.05	10	20	10	20	0	0.869		
2	<input type="checkbox"/>	2	2	0.05	10	20	10	30	0	0.968		
3	<input type="checkbox"/>	3	2	0.05	10	20	15	20	0	0.538		
4	<input type="checkbox"/>	4	2	0.05	10	20	15	30	0	0.719		
5	<input type="checkbox"/>	5	2	0.05	10	20	20	20	0	0.338		
6	<input type="checkbox"/>	6	2	0.05	10	20	20	30	0	0.478		

Create Narratives

Abbildung 6: Zweistichproben t-Test: Auswahl zur Ausgabe in Textform

1.2.2 Überlebenszeitanalyse

Als weiteres Beispiel für die Benutzung von PSS soll hier die Berechnung von Fallzahlen je Gruppe für einen zweiseitigen Log-Rank-Test bei gegebenen medianen Überlebenszeiten in den Gruppen, sowie Rekrutierungs- und Follow-up-Zeit gezeigt werden. Es werden folgende Werte in die vorgegebenen Eingabefelder eingegeben:

$\alpha = 0.05$, mediane Überlebenszeit: 24 (Gruppe 1), 36 bzw. 48 (Gruppe 2), Rekrutierungszeit: 60, Follow-up-Zeit: 60, Power: 0.80, 0.90

Bei den Zeitangaben könnte es sich hier um Monate handeln, jedoch können in den Eingabefeldern keine Einheiten angegeben werden, d.h. alle Zeitangaben müssen zur korrekten Berechnung in der gleichen Zeiteinheit angegeben werden. Die mediane Überlebenszeit für Gruppe 1 muss zweimal eingegeben werden, da immer nur Paare von medianen Überlebenszeiten erlaubt sind. Nach Drücken des Buttons „Calculate“ erhält man die in Abbildung 7 dargestellten Ergebnisse sowie die Graphik aus Abbildung 8.

Log-Rank Test for Two Survival Curves	
Fixed Scenario Elements	
Method	Lakatos normal approximation
Form of Survival Curve 1	Exponential
Form of Survival Curve 2	Exponential
Number of Sides	2
Accrual Time	60
Follow-up Time	60
Alpha	0.05
Number of Time Sub-Intervals	12
Group 1 Loss Exponential Hazard	0
Group 2 Loss Exponential Hazard	0

Computed N Per Group					
Index	Med Surv Time 1	Med Surv Time 2	Nominal Power	Actual Power	N Per Group
1	24	36	0.8	0.801	111
2	24	36	0.9	0.901	149
3	24	48	0.8	0.807	41
4	24	48	0.9	0.900	54

Abbildung 7: Fallzahlplanung Überlebenszeitanalyse: Ergebnisse

Bei der Anzeige der Ergebnisse sind auch die weiteren (Standard-)Voraussetzungen zu erkennen, die für die Fallzahlberechnung angenommen wurden. So wird die Lakatos-Normalverteilungsapproximation als Berechnungsmethode verwendet und beide Überlebenskurven werden als exponential-verteilt vorausgesetzt. In beiden Gruppen wird von keinem (Beobachtungs-)Verlust an Probanden ausgegangen.

Als beispielhafte Textausgabe erhält man folgendes:

„For a log-rank test comparing two survival curves with a two-sided significance level of 0.05, assuming uniform accrual with an accrual time of 60 and a follow-up time of 60, a sample size of 111 per group is required to obtain a power of at least 0.8 if the median survival times are 24 and 36. The actual power is 0.801.“




Abbildung 8: Überlebenszeitanalyse: graphische Ausgabe

1.3 Voreinstellungen, Konfiguration

Einige der einzugebenden Parameterwerte („Analysis Options“) sowie die gewünschten Ergebnisse („Results Options“) können unter dem Menüpunkt „Preferences“ eingestellt werden. So kann gewählt werden, ob standardmäßig Power oder Fallzahl berechnet werden soll und welche Werte für α und für die Power vorgegeben werden sollen.

Als Ergebnisse besteht die Möglichkeit, eine zusammenfassende Tabelle aller Parameter zu erhalten, eine Graphik von Power gegen Fallzahl und eine Textausgabe für ausgewählte Ergebnisse. Weiterhin kann sowohl der zu der Berechnung gehörende SAS Programmcode als auch der zugehörige SAS Log angezeigt werden. Im Programmcode wird nur die entsprechende Prozedur (POWER bzw. GLMPOWER) mit den notwendigen Befehlen angezeigt, während im Log das komplette durchgelaufene

Programm zur Berechnung und Darstellung der Ergebnisse angezeigt wird, also auch z.B. die benutzten ODS-Befehle (proc template) und die zur Ausgabe verwendeten DATA-Steps.

Alle durchgeführten Analysen werden automatisch in einem Projekt gespeichert und können daher auch später immer wieder durch Wahl des Punktes „Projects“ aufgerufen, angesehen, verändert oder gelöscht werden. Der entsprechende Auswahlbildschirm sollte analog Abbildung 9 aussehen.




Abbildung 9: Darstellung aller gespeicherten Projekte

2 Die Prozeduren POWER und GLMPOWER

Neben der interaktiven, menügesteuerten Fallzahl- bzw. Poweranalyse kann auch mit den hinter PSS liegenden SAS/STAT-Prozeduren direkt gearbeitet werden. Die ab Version 9 neuen Prozeduren POWER und GLMPOWER können an beliebigen Stellen in eigenen SAS-Programmen benutzt werden, analog zu anderen Prozeduren. Im folgenden wird die Syntax der beiden Prozeduren anhand mehrerer Beispiele vorgestellt, wobei für eine weitere und detailliertere Beschreibung der Syntax auf die SAS-Online-Hilfe [1] verwiesen werden muss. Auch eine Validierung der erhaltenen Ergebnisse kann hier nicht durchgeführt werden.

2.1 PROC POWER

Die allgemeine Syntax der Prozedur POWER, mit der eine Vielzahl von Fallzahl- und Poweranalysen durchgeführt werden können, sieht folgendermaßen aus:

```

PROC POWER < PLOTONLY > ;
  MULTREG < Optionen > ;
  ONECORR < Optionen > ;
  ONESAMPLEFREQ < Optionen > ;
  ONESAMPLEMEANS < Optionen > ;
  ONEWAYANOVA < Optionen > ;
  PAIREFREQ < Optionen > ;
  PAIREDMEANS < Optionen > ;
  TWOSAMPLEFREQ < Optionen > ;
  TWOSAMPLEMEANS < Optionen > ;
  TWOSAMPLESURVIVAL < Optionen > ;
  PLOT < Plot-Optionen > < / Graphik-Optionen > ;
RUN;

```

Alle in Klammern genannten Befehle sind optional, von den Befehlen muss einer innerhalb der Prozedur verwendet werden, der PLOT-Befehl kann nicht alleine benutzt werden. Sollen nur Graphiken erzeugt werden, gelingt dies durch Angabe der Option PLOTONLY im Prozeduraufruf. Eine Übersicht der mit den verschiedenen Befehlen möglichen Analysen findet sich in folgender Tabelle.

Kategorie	Analyse	Befehle
Mittelwerte	Ein-Stichproben t-Test	ONESAMPLEMEANS
	verbundener t-Test	PAIREDMEANS
	unverbundener t-Test	TWOSAMPLEMEANS
Konfidenzintervalle	Ein-Stichproben t-Test	ONESAMPLEMEANS
	verbundener t-Test	PAIREDMEANS
	unverbundener t-Test	TWOSAMPLEMEANS
Äquivalenztests	Ein-Stichproben t-Test	ONESAMPLEMEANS
	verbundener t-Test	PAIREDMEANS
	unverbundener t-Test	TWOSAMPLEMEANS
Anteile	eine Stichprobe	ONESAMPLEFREQ
	zwei verbundene Stichproben	PAIREFREQ
	zwei unverbundene Stichproben	TWOSAMPLEFREQ
Korrelation / Regression	Pearson-Korrelationskoeffizient	ONECORR
	multiple Regression	MULTREG
Varianzanalyse	One-way ANOVA	ONEWAYANOVA
Überlebenszeiten	Zwei-Stichproben Logrank-Test	TWOSAMPLESURVIVAL

Tabelle 1: Analysemöglichkeiten mit PROC POWER

Bei den einzelnen Befehlen können, abhängig von den bereitgestellten Analysemöglichkeiten, für die gewünschte Berechnung Angaben zum Design, zum zu verwendenden statistischen Modell und Test, zum Signifikanzniveau (α), zu den vermuteten Effekten und zur Variabilität, zur Power und zur Fallzahl eingegeben werden. Als Ausgabe erhält man dann die berechnete Fallzahl bzw. Power und evtl. zugehörige Graphiken, wenn ein PLOT-Befehl benutzt wurde. Im folgenden soll anhand einiger Beispiele ein Eindruck vermittelt werden, welche Möglichkeiten zur Power- und Fallzahlanalyse ab der SAS-Version 9 angeboten werden. Welche Optionen bei den einzelnen Befehlen innerhalb der POWER-Prozedur erlaubt sind, ist je nach gewünschten Berechnungen sehr unterschiedlich, weshalb eine vollständige Auflistung aller Optionen hier nicht sinnvoll ist.

2.1.1 Ein-Stichproben t-Test

Ein Beispiel für eine Fallzahl- bzw. Poweranalysen bei einem Ein-Stichproben t-Test sieht wie folgt aus:

```
PROC POWER ;  
  OneSampleMeans  Alpha = 0.025 0.05  Test = t  Dist = Normal  
                   Sides = 2  Mean = 10  Stddev = 5 10  
                   Power = 0.8  Ntotal = . ;  
RUN;
```

Der zu berechnende Parameter (hier die Gesamtfallzahl) wird durch Eingabe eines Missings (`Ntotal = .`) festgelegt. Bei einem t-Test unter Normalverteilungsannahme (`Test = t Dist = Normal`) kann von folgenden Parametern einer berechnet werden, während die anderen vorgegeben sein müssen: Power (`Power =`), Fallzahl (`Ntotal =`), Signifikanzniveau (`Alpha =`), Mittelwert (`Mean =`), Standardabweichung (`Stddev =`). Die Angabe, ob der Test ein- oder zweiseitig durchgeführt werden soll, ist optional, wobei zweiseitig (`Sides = 2`) voreingestellt ist, alternativ ist einseitig möglich (`Sides = 1` oder `L` oder `U` – einseitig, lower, upper). Durch weitere optionale Befehle kann der Nullmittelwert verändert werden (Default: 0 – mit `NullMean =`), die Ausgabereihenfolge der Ergebnisse festgelegt werden (mit `OutputOrder = internal | reverse | syntax`) oder die Ausgabe von exakten, nicht gerundeten Fallzahlen zu den (auf-)gerundeten Fallzahlen angefordert werden (mit `Nfractional`).

Für das angegebene Beispielprogramm wird folgende Ausgabe ins Output-Fenster geschrieben:

One-sample t Test for Mean

Fixed Scenario Elements

Distribution	Normal
Method	Exact
Number of Sides	2
Mean	10
Nominal Power	0.8
Null Mean	0

Computed N Total

Index	Std Alpha	Actual Dev	N Power	Total
1	0.025	5	0.909	6
2	0.025	10	0.835	13
3	0.050	5	0.909	5
4	0.050	10	0.803	10

Aus dieser Ausgabe lässt sich z.B. ablesen, dass bei einem einseitigen t-Test gegen 0 zum Signifikanzniveau 2,5% bei einem vermuteten Mittelwert von 10 und einer Standardabweichung von 10 dieser Effekt mit mindestens 80% Power zu entdecken ist, wenn eine Fallzahl von 13 gewählt wird, die tatsächliche Power beträgt dann 83,5%.

Entsprechende Analysen können auch durchgeführt werden, wenn statt einer Normalverteilung eine Lognormalverteilung angenommen werden soll (`Test = t` `Dist = Lognormal`). Statt der Standardabweichung muss dann der Variationskoeffizient (`CV =`) angegeben werden. Berechnet werden kann die Power oder die Fallzahl.

Neben des eigentlichen t-Tests lassen sich mit dem Befehl `OneSampleMeans` auch entsprechende Power- und Fallzahlanalysen für Konfidenzintervalle durchführen (Angabe von `ci = t` anstatt `Test = t`, nur möglich bei Normalverteilung). Hierbei sind die notwendigen Angaben das Signifikanzniveau (`Alpha =`), die vermutete halbe Breite des Intervalls (`Halfwidth =`) und die Standardabweichung (`Stddev =`). Berechnet werden kann dann entweder die Wahrscheinlichkeit bei gegebener Fallzahl die Intervallbreite zu erhalten (`Probwidth =`) oder die notwendige Fallzahl, um mit vorgegebener Wahrscheinlichkeit die Intervallbreite zu erhalten (`Ntotal =`). Der zu berechnende Parameter wird auch hier durch ein Missing festgelegt. Neben den schon

oben beim Test angegebenen optionalen Befehlen kann hier noch angegeben werden, ob die angegebene Wahrscheinlichkeit bedingt oder unbedingt interpretiert werden soll (Probtype = conditional | unconditional).

Auch hier ein Beispiel:

```
PROC POWER ;
  OneSampleMeans  Alpha = 0.025 0.05
  Ci = t  Sides = L  Stddev = 40  Halfwidth = 5
  Probwidth = 0.9  Ntotal = . ;
RUN;
```

Man erhält die folgende Ausgabe:

```
Confidence Interval for Mean

Fixed Scenario Elements

Distribution              Normal
Method                    Exact
Number of Sides           1
CI Half-Width             5
Standard Deviation        40
Nominal Prob(Width)       0.9
Prob Type                  Conditional

Computed N Total

Index  Alpha  Actual Prob (Width)  N Total
1      0.025  0.903              276
2      0.050  0.901              198
```

Um mit 90% bedingter Wahrscheinlichkeit ein einseitiges 95%-Konfidenzintervall der halben Breite 5 bei einer vermuteten Standardabweichung von 40 zu erhalten, wird eine Fallzahl von 198 benötigt.

2.1.2 Der verbundene t-Test

Für die Power- bzw. Fallzahlanalyse bei einem verbundenen t-Test sind bei Normalverteilung (Test = Diff Dist = Normal) die Befehle für das Signifikanzniveau

(Alpha =), die Standardabweichung (als gemeinsame: Stddev = oder als zwei Werte: PairedStddevs = , durch „ | “ getrennt), die Korrelation zwischen den Paaren (Corr =) und die Mittelwerte (als Differenz: MeanDiff = oder als zwei Werte: PairedMeans = , durch „ | “ getrennt) notwendig. Berechnet wird dann entweder die Power (Power =) oder die Fallzahl (Npairs =), durch Angabe von Missing gekennzeichnet. Bei Lognormalverteilung (Test = Ratio Dist = Lognormal) sind die gleichen Angaben notwendig, jedoch anstatt der Standardabweichung der Variationskoeffizient (CV = oder PairedCVs =), sowie anstatt MeanDiff = der Befehl MeanRatio = . In beiden Verteilungssituationen sind als Optionen möglich: Sides = 1 | L | U | 2, Outputorder = internal | reverse | syntax, Nfractional und Nulldiff = bzw. Nullratio = .

Ein Beispiel:

```
PROC POWER ;
  PairedMeans Alpha = 0.05 Test = Diff Dist = Normal
  Sides = 2 Stddev = 20 PairedMeans = 40 60
  Corr = 0.5 Npairs = 12 Power = . ;
```

RUN;

Ausgabe:

Paired t Test for Mean Difference

Fixed Scenario Elements

Distribution	Normal
Method	Exact
Number of Sides	2
Alpha	0.05
Mean 1	40
Mean 2	60
Standard Deviation	20
Correlation	0.5
Number of Pairs	12
Null Difference	0

Computed Power

Power

0.883

Berechnungen bei Konfidenzintervallen sind hier ebenfalls möglich.

2.1.3 Der unverbundene t-Test

Sehr ähnlich zu den bisher vorgestellten Befehlen und Optionen können auch Analysen im Zusammenhang mit einem unverbundenen Zweistichproben t-Test durchgeführt werden. Bei einem solchen t-Test unter der Annahme gleicher Varianzen in den beiden Gruppen (`Test = Diff Dist = Normal`) müssen folgende Parameter angegeben werden: Power (`Power =`), Fallzahl insgesamt (`Ntotal =`) oder Fallzahl je Gruppe (`NperGroup =`), Signifikanzniveau (`Alpha =`), Standardabweichung (`Stddev =`), sowie die Gruppenmittelwerte (`GroupMeans =`, getrennt durch „ | “) oder die Differenz dieser Mittelwerte (`MeanDiff =`). Jeder dieser Parameter (aber immer nur einer) kann durch Angabe von Missing berechnet werden. Weiterhin ist es möglich, das Verhältnis der Fallzahlen in den beiden Gruppen festzulegen (voreingestellt sind gleiche Fallzahlen), und zwar entweder über `GroupNs =` oder `GroupWeights =`, die beiden Werte jeweils durch „ | “ getrennt. Auch hier kann ein Wert durch Angabe von Missing berechnet werden.

Unter diesen Annahmen können auch Konfidenzintervalle analysiert werden (`Ci = Diff Dist = Normal`).

Bei einem t-Test unter der Annahme ungleicher Varianzen in den beiden Gruppen (`Test = Diff_Satt Dist = Normal`) sind folgende Parameter möglich: Power (`Power =`), Fallzahl insgesamt (`Ntotal =`) oder Fallzahl je Gruppe (`NperGroup =`), Signifikanzniveau (`Alpha =`), Standardabweichungen in den beiden Gruppen (`GroupStddevs =`, getrennt durch „ | “), sowie die Gruppenmittelwerte (`GroupMeans =`, getrennt durch „ | “) oder die Differenz dieser Mittelwerte (`MeanDiff =`). Berechnet werden können hier nur Power oder Fallzahl. Gruppengewichte können nicht angegeben werden.

Schließlich gibt es noch die Möglichkeit eines t-Tests unter Annahme einer Lognormalverteilung (`Test = Ratio Dist = Lognormal`) mit den Parametern Variationskoeffizient (`CV =`, evtl. durch „ | “ getrennt) anstatt Standardabweichung und evtl. dem Befehl `MeanRatio =` anstatt `MeanDiff =`. Hier werden Power oder Fallzahl berechnet.

Bei allen drei Möglichkeiten können als weitere Optionen (wie schon in den Situationen zuvor) die Befehle `Sides =`, `Outputorder =`, `Nfractional` und `NullDiff =` bzw. `NullRatio =` verwendet werden.

Zum unverbundenen t-Test ein Beispiel, in dem die Fallzahl in einer der beiden Gruppen berechnet werden soll:

```

PROC POWER ;
  TwoSampleMeans Alpha = 0.025 Test = Diff Dist = Normal
  Sides = U GroupMeans = 2 | 3 3.2
  Stddev = 3 Power = 0.8 Groups = 100 | . ;
RUN;

```

Die Ergebnisse lauten:

Two-sample t Test for Mean Difference

Fixed Scenario Elements

Distribution	Normal
Method	Exact
Number of Sides	U
Alpha	0.025
Group 1 Mean	2
Standard Deviation	3
Group 1 Sample Size	100
Nominal Power	0.8
Null Difference	0

Computed N2

Index	Mean2	Actual Power	N2
1	3.0	0.800	246
2	3.2	0.802	99

2.1.4 Äquivalenztests

Analysen im Zusammenhang mit Äquivalenztests sind für Einstichproben-Mittelwerte (Test = Equiv Dist = Normal | Lognormal – OneSampleMeans), verbundene Mittelwerte (Test = Equiv_Diff Dist = Normal oder Test = Equiv_Ratio Dist = Lognormal – PairedMeans) und unverbundene Mittelwerte (Test = Equiv_Diff Dist = Normal oder Test = Equiv_Ratio Dist = Lognormal – TwoSampleMeans) möglich. In allen Fällen können entweder Power (Power =) oder die Fallzahl (Ntotal = bei OneSampleMeans, Npairs = bei PairedMeans, Ntotal = oder NperGroup = bei TwoSampleMeans) berechnet werden. Des weiteren muss das Signifikanzniveau (Alpha =), die Standardabweichung (Stddev = oder PairedStddevs = bei PairedMeans) bzw. der Variationskoeffizient (CV = oder PairedCVs = bei PairedMeans), die untere (Lower =) und obere

(Upper =) Grenze des Äquivalenzbereichs, sowie der bzw. die Mittelwerte (Mean = bei OneSampleMeans, MeanDiff = oder PairedMeans = oder MeanRatio = bei OneSampleMeans, GroupMeans = oder MeanDiff = oder MeanRatio = bei TwoSampleMeans) angegeben werden. Zusätzlich sind die Optionen Outputorder = und Nfractional möglich. Wann welche Befehle möglich sind, ist völlig analog zu den entsprechenden vorher beschriebenen klassischen Testsituationen festgelegt. Abschließend hier noch ein Beispiel für zwei unverbundene Stichproben bei Normalverteilung:

```
PROC POWER ;
  TwoSampleMeans Alpha = 0.05 Test = Equiv_Diff Dist = Normal
  Lower = 2 Upper = 5 MeanDiff = 4
  Stddev = 8 Power = 0.9 Ntotal = . ;
RUN;
```

Man erhält folgende Ausgabe:

Equivalence Test for Mean Difference

Fixed Scenario Elements

Distribution	Normal
Method	Exact
Lower Equivalence Bound	2
Upper Equivalence Bound	5
Alpha	0.05
Mean Difference	4
Standard Deviation	8
Nominal Power	0.9
Group 1 Weight	1
Group 2 Weight	1

Computed N Total

Actual	N
Power	Total
0.900	2194

2.1.5 Anteile – eine Stichprobe

Für Fallzahl- bzw. Poweranalysen bei Anteilen in einer Stichprobe kann die Berechnung exakt (Test = AdjZ | Z Method = Exact oder Test = Exact) oder approximativ (Test = AdjZ | Z Method = Normal) durchgeführt werden. Bei exakten Berechnungen kann nur die Power berechnet werden, bei approximativer Berechnung Fallzahl oder Power (Ntotal = , Power =). Notwendige Angaben sind: Signifikanzniveau (Alpha =), Anteil (Proportion =) und Null-Anteil (NullProportion =). Als weitere Optionen können Sides = , Outputorder = oder Nfractional verwendet werden.

Ein Beispiel:

```
PROC POWER ;
  OneSampleFreq Alpha = 0.05 Test = Z Method = Normal
  Proportion = 0.25 NullProportion = 0.2
  Power = 0.8 0.9 Ntotal = . ;
RUN;
```

Ergebnis:

Z Test for Binomial Proportion

Fixed Scenario Elements

Method	Normal approximation
Null Proportion	0.2
Alpha	0.05
Binomial Proportion	0.25
Number of Sides	2

Computed N Total

Index	Nominal Power	Actual Power	N Total
1	0.8	0.800	528
2	0.9	0.900	718

2.1.6 Anteile – zwei verbundene Stichproben

Auch bei zwei verbundenen Stichproben können Analysen basierend auf dem Test von McNemar sowohl exakt (Test = McNemar Method = Exact Dist = Exact_Cond | Normal) als auch approximativ (Test = McNemar Method = Connor | Miettinen Dist = Normal) durchgeführt werden, wobei bei den exakten Methoden nur die Power berechnet werden kann (Npairs = , Power =). Neben dem Signifikanzniveau (Alpha =) müssen die relativen Häufigkeiten der beiden diskordanten Zellen p_{10} und p_{01} angegeben werden. Dies kann durch eine eindeutige Auswahl aus folgenden Befehlen geschehen: DiscProportions = (getrennt durch „ | “), DiscPropDiff = , TotalPropDisc = , DiscPropRatio = , RefProportion = . Als weitere optionale Befehle können wie schon zuvor beschrieben Sides = , Outputorder = , Nfractional und auch NullDiscPropRatio = (voreingestellt = 1) verwendet werden.

Auch hier soll ein Beispiel die Anwendung erläutern:

```
PROC POWER ;
    PairedFreq Alpha = 0.05 Test = McNemar Method = Exact
    Dist = Exact_Cond Sides = 2 DiscProportions = 0.15 | 0.25
    Npairs = 250 Power = . ;
RUN;
```

Die berechnete Power ist aus folgender Ausgabe abzulesen:

```

                McNemar Exact Conditional Test

                Fixed Scenario Elements

Distribution                               Exact conditional
Method                                       Exact
Nominal Alpha                               0.05
Success-Failure Proportion                 0.15
Failure-Success Proportion                 0.25
Number of Pairs                             250
Number of Sides                             2
Null Ratio of Discordant Proportions       1

                Computed Power

                Alpha    Actual
                Power    Power

                0.0395   0.677
```

2.1.7 Anteile – zwei unverbundene Stichproben

Fallzahl- und Poweranalysen für Anteile in zwei unverbundenen Stichproben (Vierfeldertafel) sind auf Basis des exakten Fisher-Tests, des Pearson- χ^2 -Tests oder des Likelihood-Ratio- χ^2 -Tests durchzuführen (`Test = Fisher | Lrchi | Pchi`). Es können jeweils die Power (`Power =`) oder die Fallzahl (`Ntotal =` oder `NperGroup =` oder `Groupns =`) berechnet werden. Neben dem Signifikanzniveau (`Alpha =`) können die relativen Häufigkeiten in den beiden Gruppen durch eindeutige Kombination aus folgenden Befehlen angegeben werden: `GroupProportions =` (getrennt durch „ | “), `NullProportionDiff =` (voreingestellt = 0), `OddsRatio =`, `RefProportion =`, `RelativeRisk =`. Als weitere optionalen Befehle sind neben den schon bekannten `Sides =`, `Outputorder =` und `Nfractional` auch noch folgende möglich: `GroupWeights =` (Festlegung des Verhältnisses der Fallzahlen, Werte durch „ | “ getrennt), `NullOddsRatio =` und `NullRelativeRisk =` (jeweils voreingestellt mit = 1).

Abschließend ein Beispiel:

```
PROC POWER ;
  TwoSampleFreq Alpha = 0.05 Test = Pchi Sides = 2
  OddsRatio = 2.5 RefProportion = 0.3 GroupWeights = (1 3)
  Power = 0.8 Ntotal = . ;
RUN;
```

Es erscheint die folgende Ausgabe:

```
Pearson Chi-square Test for Two Proportions

Fixed Scenario Elements

Distribution           Asymptotic normal
Method                Normal approximation
Number of Sides      2
Alpha                 0.05
Reference (Group 1) Proportion 0.3
Odds Ratio            2.5
Group 1 Weight       1
Group 2 Weight       3
Nominal Power        0.8
Null Odds Ratio      1
```

```
Computed N Total
      Actual      N
      Power      Total
      0.806      216
```

2.1.8 One-Way ANOVA

Für eine One-Way ANOVA (Varianzanalyse) können Fallzahl- und Powerberechnungen sowohl für den Overall-Test (`Test = Overall`) als auch für einzelne Kontraste (`Test = Contrast`) durchgeführt werden. Es muss in beiden Fällen das Signifikanzniveau (`Alpha =`), die Mittelwerte der einzelnen Gruppen (`GroupMeans =`, jeweils durch „ | “ getrennt) und die Standardabweichung (`Stddev =`) angegeben werden. Wenn Kontraste untersucht werden sollen, müssen solche mit `Contrast =` angegeben werden. Berechnet werden kann entweder die Power (`Power =`) oder die Fallzahl (`Ntotal =`, `NperGroup =`, `Groupns =`). Als zusätzliche Optionen sind in beiden Fällen möglich: `Outputorder =`, `Nfractional` und `GroupWeights =` (getrennt durch „ | “). Bei Kontrasten kann auch `Sides =` und `NullContrast =` (voreingestellt = 0) verwendet werden.

Im folgenden zu beiden Fällen je ein Beispiel:

```
PROC POWER ;
  OneWayAnova  Alpha = 0.05  Test = Overall
  GroupMeans = 4 | 6 | 7 | 8  Stddev = 4 5
  Power = 0.8  NperGroup = . ;
RUN;
```

Ausgabe:

```
Overall F Test for One-Way ANOVA

      Fixed Scenario Elements

      Method              Exact
      Alpha                0.05
      Group Means          4 6 7 8
      Nominal Power        0.8
```

Computed N Per Group			
Index	Std Dev	Actual Power	N Per Group
1	4	0.801	21
2	5	0.812	33

Beispiel für Kontraste:

```
PROC POWER ;
  OneWayAnova  Alpha = 0.05  Test = Contrast
  GroupMeans = 4 | 6 | 7 | 8
  Contrast = (1 -1 0 0) (0 0 1 -1)
  Stddev = 4 5  Power = 0.8  NperGroup = . ;
RUN;
```

Ausgabe:

Single DF Contrast in One-Way ANOVA

Fixed Scenario Elements

Method	Exact
Alpha	0.05
Group Means	4 6 7 8
Nominal Power	0.8
Number of Sides	2
Null Contrast Value	0

Computed N Per Group

Index	-----Contrast-----				Std Dev	Actual Power	N Per Group
1	1	-1	0	0	4	0.804	64
2	1	-1	0	0	5	0.802	99
3	0	0	1	-1	4	0.801	252
4	0	0	1	-1	5	0.800	393

2.1.9 Korrelation und Regression

Für Analysen in Zusammenhang mit dem Pearson-Korrelationskoeffizienten (Test = Pearson Dist = FisherZ oder Test = Pearson Dist = T Model = fixed

| random) wird bei Angabe des Signifikanzniveaus (Alpha =) und des Korrelationskoeffizienten (Corr =) entweder die Power (Power =) oder die Fallzahl (Ntotal =) berechnet. Neben den schon bekannten Optionen Sides = , Outputorder = und Nfractional kann auch die im Test zu verwendende Nullkorrelation mit NullCorr = angegeben werden. Die Anzahl evtl. zu berücksichtigender zusätzlicher Variablen (partielle Korrelation) kann mit NpartialVars = angegeben werden.

Beispiel:

```
PROC POWER ;
    OneCorr Alpha = 0.05 Test = Pearson Dist = FisherZ
    NullCorr = 0.5 Corr = 0.6 0.7 Power = 0.9 Ntotal = . ;
RUN;
```

Ausgabe:

```
Fisher's z Test for Pearson Correlation

Fixed Scenario Elements

Distribution          Fisher's z transformation of r
Method                Normal approximation
Null Correlation      0.5
Nominal Alpha         0.05
Nominal Power         0.9
Number of Sides       2
Number of Variables Partialled Out 0

Computed N Total

Index   Corr   Actual Alpha   Actual Power   N Total
1       0.6   0.0500      0.901         511
2       0.7   0.0499      0.902         107
```

Auch Analysen für multiple lineare Regressionsmodelle sind möglich (MultReg Test = Type3). Neben der Angabe von α (Alpha =) und der Modellart (Model = fixed | random) muss zur Konkretisierung des Modells aus folgenden Befehlen gewählt werden: NFullPredictors = , NReducedPredictors = , NTestPredictors = , PartialCorr = , RsquareDiff = , RsquareFull = , RsquareReduced = . Es können dann Fallzahl oder Power berechnet werden (Ntotal = , Power =). Optional

kann `Outputorder =`, `Nfractional` und `Noint` (kein Absolutglied im Regressionsmodell) verwendet werden.

Auch hier noch ein Beispiel:

```
PROC POWER ;
  MultReg Alpha = 0.05 Test = Type3 Model = random
  NFullPredictors = 3 NTestPredictors = 1
  PartialCorr = 0.25 Power = 0.8 Ntotal = . ;
RUN;
```

Ausgabe:

Type III F Test in Multiple Regression

Fixed Scenario Elements

Method	Exact
Model	Random X
Number of Predictors in Full Model	3
Number of Test Predictors	1
Alpha	0.05
Partial Correlation	0.25
Nominal Power	0.8

Computed N Total

Actual Power	N Total
0.802	125

2.1.10 Überlebenszeitanalysen

Im Zusammenhang mit Überlebenszeitanalysen gibt es eine Vielzahl von Möglichkeiten zur Fallzahl- oder Powerberechnung. Es stehen drei verschiedene Testverfahren zur Verfügung (`Test = Gehan | Logrank | TaroneWare`) und es kann in allen Situationen entweder die Power (`Power =`) oder die Fallzahl (`Ntotal =` , `NperGroup =` , `Groupns =`) berechnet werden. Angegeben werden müssen das Signifikanzniveau (`Alpha =`), Informationen zur Rekrutierungs- und Follow-Up-Zeit (`AccrualTime =` , `FollowupTime =` , `TotalTime =`) und Informationen zum mittleren Überleben (`GroupMedSurvTimes =` , `GroupSurvExpHazards =` , bei beiden Werte jeweils getrennt durch „ | “, `RefSurvExpHazard =` , `HazardRatio =` . Die

Überlebenskurven werden standardmäßig als exponentialverteilt angenommen, es sei denn, eine Kurve wird durch Angabe von mehr als einer Überlebenswahrscheinlichkeit zu bestimmten Zeiten definiert, in diesen Fällen wird automatisch eine stückweise lineare Überlebenskurve angenommen. Die Definition einer solchen Kurve geschieht mit dem Befehl `Curve = ti : pi` (t_i : Zeitpunkt, p_i : Überlebenswahrscheinlichkeit). Dies wird in unten stehendem Beispiel verdeutlicht. Als Optionen sind neben den üblichen `Sides =`, `Outputorder =`, `Nfractional` auch `GroupWeights =` (Gruppengewichte), `NsubIntervals =` (Anzahl Subintervalle je Zeiteinheit für die Berechnungen – voreingestellt = 12) und `GroupLoss =`, `GroupMedLossTime =`, `GroupLossExpHazards =` (für Beobachtungsverluste).

In folgendem Beispiel soll die Fallzahl je Gruppe bei einem Logrank-Test bestimmt werden, wenn eine Überlebenskurve stückweise linear definiert wird mit Überlebenswahrscheinlichkeiten von 0.8, 0.6 und 0.5 nach 1, 2 bzw. 3 Jahren und die andere Überlebenskurve exponentialverteilt mit einer Überlebenswahrscheinlichkeit von 0.5 nach 2 Jahren. Die Rekrutierungszeit soll 2 Jahre, die Follow-Up-Zeit 1 Jahr betragen. Das zugehörige Programm lautet:

```
PROC POWER ;
  TwoSampleSurvival Alpha = 0.05 Test = Logrank
  GroupSurvival = "A" | "B"
  Curve("A") = (1 2 3) : (0.8 0.6 0.5)
  Curve("B") = 2 : 0.5
  AccrualTime = 2 FollowupTime = 1
  Power = 0.8 NperGroup = . ;
RUN;
```

In folgender Ausgabe kann man eine Fallzahl von 239 je Gruppe ablesen:

```

Log-Rank Test for Two Survival Curves

Fixed Scenario Elements

Method                      Lakatos normal approximation
Accrual Time                  2
Follow-up Time                1
Alpha                        0.05
Group 1 Survival Curve       A
Form of Survival Curve 1     Piecewise Linear
Group 2 Survival Curve       B
Form of Survival Curve 2     Exponential
Nominal Power                 0.8
Number of Sides               2
Number of Time Sub-Intervals 12
Group 1 Loss Exponential Hazard 0
Group 2 Loss Exponential Hazard 0

Computed N Per Group
Actual      N Per
Power      Group
0.801      239

```

2.2 Graphiken mit PROC POWER

Zu jeder möglichen Fallzahl- oder Poweranalyse, die in den vorigen Abschnitten vorgestellt wurde, kann der Zusammenhang zwischen Fallzahl und Power graphisch dargestellt werden. Durch Angabe der Option `PLOTONLY` hinter dem Prozeduraufruf wird nur die Graphik und nicht die entsprechende Textausgabe erzeugt. Der Befehl zur Graphikerzeugung lautet

```
Plot <Plot-Optionen> </Graphik-Optionen>;
```

mit den folgenden möglichen Optionen:

Plot-Optionen:

```

Interpol = join | none
Key = ByCurve <(numbers=on|off pos=bottom|inset)> |
      ByFeature <(pos=bottom|inset)> | OnCurves
Markers = analysis | computed | nice | none
Max = max          Min = min          NPoints = n          Step = n

```

R. Minkenberg

```
VarY (Color|Linestyle|Panel|Symbol <by Paramater> ...)  
X = effect | n | power          Y = effect | n | power  
Xopts | Yopts = (CrossRef=no | yes  Ref= $n_1$   $n_2$  ...)
```

Graphik-Optionen:

```
Description = "Name"           Name = "Name"
```

Mit zwei Beispielprogrammen und den zugehörigen Graphiken soll deutlich werden, welche Ausgaben möglich sind.

Beispiel 1:

```
PROC POWER plotonly ;  
  TwoSampleMeans Alpha = 0.025 Test = Diff Dist = Normal  
  Sides = U GroupMeans = (1 1.5 2) | (2.5 4 5.5)  
  Stddev = 5 to 7 by 0.2 Power = 0.8 Groupns = 100 | . ;  
  Plot X = Groupns Key = OnCurves  
  Interpol=none Yopts = (ref=1e3 2e3 5e3)  
  VarX (Linestyle by GroupMeans Symbol by Stddev) ;  
RUN;
```

Ausgabe:




Abbildung 10: Graphikausgabe für Beispiel 1

Beispiel 2:

```

PROC POWER plotonly;
  TwoSampleFreq Alpha = 0.05 Test = Pchi Sides = 2
  OddsRatio = 2.5 to 3 by 0.1 RefProportion = 0.3
  GroupWeights = (1 3) Power = 0.8 0.9 Ntotal = . ;
  Plot Y = Power VarY (Color by OddsRatio)
      Min = 0.5 Max = 1 Key = ByCurve ;
RUN;

```

Ausgabe:




Abbildung 11: Graphikausgabe für Beispiel 2

2.3 PROC GLMPOWER

Innerhalb von PROC POWER ist bei Varianzanalysen nur eine One-Way ANOVA möglich. Für komplexere varianzanalytische Modelle steht PROC GLMPOWER zur Verfügung. Mit dieser Prozedur kann sowohl Power als auch Fallzahl für nahezu beliebige Modelle bestimmt werden und zwar sowohl für globale Tests als auch zum Testen von Kontrasten. Die Syntax von GLMPOWER ist stark angelehnt an die bekannte Syntax der Prozedur GLM:

```

PROC GLMPOWER < DATA=Datei PLOTONLY > ;
  CLASS Variablen ;
  MODEL YVars = Effekte ;
  WEIGHT Variable ;
  CONTRAST "Label" Effekt Werte </ Singular = n> ;
  POWER Optionen ;
  PLOT < Plot-Optionen > </ Graphik-Optionen > ;
RUN;

```

Die Befehle CLASS, MODEL, WEIGHT und CONTRAST werden analog zur GLM-Prozedur verwendet. Die direkte Berücksichtigung von stetigen Kovariablen ist nicht möglich. Der Befehl POWER erlaubt folgende Optionen:

```

Alpha =      Ntotal =      Power =      Stddev =      Nfractional
Outputorder = internal | reverse | syntax
CorrXY =      NCovariates =      PropVarReduction =

```

Eine Berücksichtigung von Kovariablen ist durch Angabe der Anzahl an solchen (NCovariates =) indirekt möglich. Die anderen Optionen sind bekannt, die zu berechnende Größe wird wieder durch Missing codiert. Die einzelnen Gruppenmittelwerte und eventuelle Gewichte bzgl. der Fallzahl müssen vor Aufruf von GLMPOWER in einer SAS-Datei gespeichert sein.

In folgendem Beispiel sollen Fallzahlen für die globalen Tests und für drei Kontraste bestimmt werden. Die vermuteten Mittelwerte je Gruppe und die Fallzahlverhältnisse zwischen den Gruppen sind folgenden Tabellen zu entnehmen:

	Behandlung				
Geschlecht	0	A	B	C	D
männl.	40	38.5	31	36	27.5
weibl.	38	34	28.5	33	24

Tabelle 2: Gruppenmittelwerte für GLMPOWER-Beispiel

	Behandlung				
Geschlecht	0	A	B	C	D
männl.	2	3	4	3	3
weibl.	3	6	8	6	6

Tabelle 3: Fallzahlgewichte für GLMPOWER-Beispiel

R. Minkenberg

Das zugehörige Beispielprogramm lautet, wenn eine Kovariable mit Korrelationskoeffizienten von 0, 0.2 und 0.4 zur abhängigen Variablen mitberücksichtigt wird:

```
DATA beispiel ;
  input Behandlg $ Sex $ Wert CellWgt @@ ;
  datalines ;
  0 m 40 2 0 w 38 3
  A m 38.5 3 A w 34 6
  B m 31 4 B w 28.5 8
  C m 36 3 C w 33 6
  D m 27.5 3 D w 24 6
  ;
RUN ;

PROC GLMPOWER Data = beispiel ;
  Class Behandlg Sex ;
  Model Wert = Behandlg Sex ;
  Weight CellWgt ;
  Contrast "0 vs. others" Behandlg 4 -1 -1 -1 -1 ;
  Contrast "AB vs. CD" Behandlg 0 1 1 -1 -1 ;
  Contrast "A vs. D" Behandlg 0 1 0 0 -1 ;
  Power Stddev = 10 NCovariates = 1
  CorrXY = 0 0.2 0.4 Alpha = 0.05 Power = 0.8 Ntotal = . ;
RUN ;
```

Man erhält folgende Ausgabe, in der für alle globalen Tests und alle verlangten Kontraste die Fallzahlen angegeben sind:

Index	Type	Source	Corr XY	Std Dev	Test DF	Error DF	Actual Power	N Total
1	Effect	Behandlg	0.0	10.00	4	81	0.918	88
2	Effect	Behandlg	0.2	9.80	4	81	0.929	88
3	Effect	Behandlg	0.4	9.17	4	81	0.959	88
4	Effect	Sex	0.0	10.00	1	389	0.841	396
5	Effect	Sex	0.2	9.80	1	345	0.813	352
6	Effect	Sex	0.4	9.17	1	301	0.812	308
7	Contrast	0 vs. others	0.0	10.00	1	169	0.887	176
8	Contrast	0 vs. others	0.2	9.80	1	169	0.899	176
9	Contrast	0 vs. others	0.4	9.17	1	125	0.849	132
10	Contrast	AB vs. CD	0.0	10.00	1	477	0.828	484
11	Contrast	AB vs. CD	0.2	9.80	1	433	0.807	440
12	Contrast	AB vs. CD	0.4	9.17	1	389	0.818	396
13	Contrast	A vs. D	0.0	10.00	1	81	0.865	88
14	Contrast	A vs. D	0.2	9.80	1	81	0.878	88
15	Contrast	A vs. D	0.4	9.17	1	81	0.917	88

3 Zusammenfassung

In der Version 9 des SAS-Systems ist die schon lange erhoffte Einfügung von Fallzahl- und Poweranalysen mit den Prozeduren POWER und GLMPOWER gut gelungen. Neben diesen SAS/STAT-Prozeduren, die eine Vielzahl an Möglichkeiten der Text- und Graphikausgabe ermöglichen, wird auch eine interaktive Oberfläche (PSS) bereitgestellt, mittels der aus jedem Internetbrowser heraus leicht bedienbar und verständlich die vorhandenen Fallzahl- bzw. Poweranalysen auch ohne detaillierte Kenntnis der zugrundeliegenden Prozeduren durchgeführt werden kann. Leider kann PSS mit der „deutschen“ SAS-Version (noch) nicht benutzt werden.

Die vorhandenen Analysemöglichkeiten decken die üblichen und häufigsten Situationen bei der Fallzahlplanung ab, wobei in einzelnen Fällen Ergänzungen sicher möglich sind. Damit in Zukunft die neu geschaffenen Möglichkeiten tatsächlich akzeptiert und dann auch benutzt werden, ist eine Validierung der berechneten Ergebnisse notwendig. Hierbei könnte helfen, dass sehr viele Ergebnisse nach den ersten durchgeführten Berechnungen Ähnlichkeit mit den Ergebnissen des bekannten Fallzahlanalyseprogramms nQuery (Statistical Solutions) zu haben scheinen.

Literatur

- [1] SAS Online Documentation, Version 9,
<http://support.sas.com/documentation/onlinedoc/sas9doc.html>
- [2] nQuery Advisor, <http://www.statsol.ie/nquery/nquery.htm>